```
Maxima 5.13.0 http://maxima.sourceforge.net
Using Lisp GNU Common Lisp (GCL) GCL 2.6.7 (aka GCL)
Distributed under the GNU Public License. See the file COPYING.
Dedicated to the memory of William Schelter.
This is a development version of Maxima. The function bug_report()
provides bug reporting information.
```

```
(%i37) kill(all)$
```

This document recreates the Maple document from Honor Calculus II using free software of Maxima and texmacs. Note that Maxima does not rewrite the integrals in terms of elliptic functions as does Maple.

```
(%i1) f(x):=sqrt(cos(x)^2+1);
```

(%o1) $f(x):=\sqrt{\cos(x)^2+1}$

```
(%i2) integrate(f(x),x,0,%pi);
```

(%o2) $\int_0^{\pi}\sqrt{\cos(x)^2+1}\,\mathrm{d}\,x$

Making the trigonometric substitution $\cos^2 x = 1 - \sin^2 x$ using symmetry transforms the integral into a complete elliptic integral of the second kind.

```
(%i3) 2*sqrt(2.0)*elliptic_e(%pi/2,0.5);
```

(%o3) 3.820197789027714

This implements double precision floating point quadrature using Fortran routines that have been machine translated into the Maxima language.

```
(%i4) quad_qags(f(x),x,0,%pi);
```

(%o4) $[3.820197789027712, 1.301576854452153\times 10^{-13}, 63, 0]$

For high precision integration we use the bigfloat implementation of the Romberg integrator. The computation is done with 60 digits and an error of 10^{-40}. The result is stored in the variable a and then converted to a 40 digit bigfloat in which all digits are correct. The computation takes significantly longer than with Maple which recognizes the Elliptic integral.

```
(%i5) load(brmbrg);
```

(%o5) /usr/share/maxima/5.13.0/share/numeric/brmbrg.lisp

```
(%i6) a:(fpprec:60,
        brombergtol: 0.0b0,
        brombergabs:1.0b-40,
        brombergit: 20,
        bromberg(f(x),x,0,%pi));
```

(%o6) $3.82019778902771201790476208217144329190996761464727472675652_B\times 10^0$

```
(%i7) fpprec:40;
```

(%o7) 40

```
(%i8) bfloat(a);
```

(%o8) $3.820197789027712017904762082171443291 91_B\times 10^0$

We now compute this Elliptic integral quickly using the arithemetic-geometric mean iteration of Gauss and Legendre in the way the Maple probably did it.

Define $a_{n+1} = (a_n + b_n)/2$, $b_{n+1} = \sqrt{a_n b_n}$ and $c_{n+1} = (a_n - b_n)/2$ with $a_0 = 1$, $b_0 = \sqrt{1-k^2}$ and $c_0 = k$. Then $K = \lim_{n\to\infty} a_n$ and $(K - E)/K = (1/2)\sum_{n=0}^{\infty} 2^n c_n^2$ where K and E are the complete elliptic integrals of the first and second kind respectively.

```
(%i9) myE(k):=block([E,K,xn,n,bfzero],
          local(phi),
          xn:[1,sqrt(1-k^2),k],
          phi(x):=bfloat([(x[1]+x[2])/2,sqrt(x[1]*x[2]),1/2*(x[1]-x[2])]),
          E:0, n:0, bfzero:bfloat(0),
          while xn[3]#bfzero do
              (E:E+2^n*xn[3]^2, n:n+1, xn:phi(xn)),
          E:E+2^n*xn[3]^2,
          K:bfloat(%pi/xn[1]/2),
          E:(2-E)*K/2,
          return(E))$
(%i10) bfloat(2*sqrt(2)*myE(1/sqrt(2)));
```

(%o10) $3.820197789027712017904762082171443329191_B \times 10^0$

```
(%i11) bfloat(%pi);
```

(%o11) $3.141592653589793238462643383279502884197_B \times 10^0$

```
(%i12) integrate(x,x);
```

(%o12) $\frac{x^2}{2}$

```
(%i13) integrate(x,x,1,14);
```

(%o14) $\frac{195}{2}$

```
(%i15)
```